

# SAS<sup>®</sup> GLOBAL FORUM 2018

---

USERS PROGRAM

## Tweaking your Tables

Suppressing superfluous  
subtotals in PROC TABULATE

April 8 - 11 | Denver, CO

**#SASGF**

# Tweaking your Tables

Steve Cavill

Infoclarify

## Introduction

PROC TABULATE is a great tool for generating cross tab style reports. It's very flexible but has a few little limitations. One is suppressing superfluous subtotals. The ALL keyword creates a total or subtotal for the categories in one dimension. However, if there is only one category in the dimension, the subtotal is still shown, which is really just repeating the detail line again. This can look a bit confusing in the final output. This talk demonstrates a method to suppress those superfluous totals by saving the output from PROC TABULATE using the OUT= option. That data set is then reprocessed to remove the undesirable totals using the \_TYPE\_ variable which identifies the total rows. PROC TABULATE is then run again against the reprocessed data set to create the final table.

The technique highlights the flexibility of the SAS® programming language to get exactly the output you want.

## Example PROC TABULATE output

			N	MSRP Mean
Origin	Make	Type		
Asia	Isuzu	SUV	2	\$26,149
		All	2	\$26,149
	All		2	\$26,149
USA	Make	Type		
	Ford	SUV	4	\$32,055
		Sedan	11	\$20,835
		Sports	3	\$28,418
		Truck	3	\$23,312
		Wagon	2	\$19,883
	All		23	\$24,018
	Jeep	Type		
		SUV	3	\$24,518
		All	3	\$24,518
	All		26	\$24,074
Grand Total			28	\$24,222

Example uses subset of SASHELP.CARS data set

```
proc tabulate data=data format=6.;
class origin make type ;
var msrp;
tables origin
      *(make*(type all) all)
      all='Grand Total'
      ,n msrp*mean*f=dollar7.;
```

## Desired output:

- The extra total lines in Figure 1 are superfluous
- Suppress them and add data specific subtotal headings, as in Fig 2

			N	MSRP Mean
Origin	Make	Type		
Asia	Isuzu	SUV	2	\$26,149
		All	2	\$26,149
	All		2	\$26,149
USA	Make	Type		
	Ford	SUV	4	\$32,055
		Sedan	11	\$20,835
		Sports	3	\$28,418
		Truck	3	\$23,312
		Wagon	2	\$19,883
	All		23	\$24,018
	Jeep	Type		
		SUV	3	\$24,518
		All	3	\$24,518
	All		26	\$24,074
Grand Total			28	\$24,222

Figure-1: Standard-output

			N	MSRP_Mean
			Sum	Sum
origin	make	type		
Asia	Isuzu	SUV	2	\$26,149
USA	Ford	SUV	4	\$32,055
		Sedan	11	\$20,835
		Sports	3	\$28,418
		Truck	3	\$23,312
		Wagon	2	\$19,883
	Ford Total		23	\$24,018
	Jeep	SUV	3	\$24,518
	USA Total		26	\$24,074
Grand Total			28	\$24,222

Figure-2: Modified-output

## Method – Step 1 send output to a table

- Send output to a table:
 

```
proc tabulate
data=data
out=table...
```
- \_TYPE\_
  - 111 = detail
  - 110 = Make total
  - 100 = Origin total
  - 000 = Grand total
- We want to delete the marked rows

	Origin	Make	Type	_TYPE_	N	MSRP_Mean
1	Asia	Isuzu	SUV	111	2	26149
2	USA	Ford	SUV	111	4	32055
3	USA	Ford	Sedan	111	11	20835.454545
4	USA	Ford	Sports	111	3	28418.333333
5	USA	Ford	Truck	111	3	23311.666667
6	USA	Ford	Wagon	111	2	19882.5
7	USA	Jeep	SUV	111	3	24518.333333
8	Asia	Isuzu		110	2	26149
9	USA	Ford		110	23	24015.869565
10	USA	Jeep		110	3	24518.333333
11	Asia			100	2	26149
12	USA			100	26	24073.846154
13				000	28	24222.071429

## Method – Step 2 remove the subtotals we don't want

- Which rows to delete?
  - \_type\_=110 (Make total) and one type per make, OR
  - \_type\_=100 (Origin total) and one make per origin

- Identify rows to delete

```
proc sql;
create table make_suppress as
select make,count (distinct type) as freq
from data
group by make
having count (distinct type)=1
; /* only one type within the make*/
```

Make	Freq
Isuzu	1
Jeep	1

```
create table origin_suppress as
select origin,count (distinct make) as freq
from data
group by origin
having count (distinct make)=1
; /* only one make in the origin */
```

Origin	Freq
Asia	1

```
/* now delete the rows */
Proc sql;
delete from table where _type_='110'
and make in (select make from make_suppress);
/* will delete Jeep and Isuzu totals */
delete from table where _type_='100'
and origin in (select origin from origin_suppress)
; /* will delete Asia subtotal */
;
```

# Tweaking your Tables


Steve Cavill

Infoclarify

## Method – Step 3 Make data specific headings

```
data table;
  length origin make type $20; /* increase label width*/
  set table;
/* first change the total labels (which are blank in the output
data set */
  if _type_='000' then origin='Grand Total' ; /*grand total*/
/*origin total - include the value of Origin on the total line */
  if _type_='100' then make=catx(' ',origin,'Total') ;
/*make total-include the value of Make on the total line */
  if _type_='110' then type=catx(' ',make,'Total') ;
run;
```

	Origin	Make	Type	_TYPE_	N	MSRP_Mean
1	Asia	Isuzu	SUV	111	2	26149
2	USA	Ford	SUV	111	4	32055
3	USA	Ford	Sedan	111	11	20835.454545
4	USA	Ford	Sports	111	3	28418.333333
5	USA	Ford	Truck	111	3	23311.666667
6	USA	Ford	Wagon	111	2	19882.5
7	USA	Jeep	SUV	111	3	24518.333333
8	Asia	Isuzu		110	2	26149
9	USA	Ford		110	23	24015.869565
10	USA	Jeep		110	3	24518.333333
11	Asia			100	2	26149
12	USA			100	26	24073.846154
13				000	28	24222.071429



	origin	make	type	_TYPE_	N	MSRP_Mean
1	Asia	Isuzu	SUV	111	2	26149
2	USA	Ford	SUV	111	4	32055
3	USA	Ford	Sedan	111	11	20835.454545
4	USA	Ford	Sports	111	3	28418.333333
5	USA	Ford	Truck	111	3	23311.666667
6	USA	Ford	Wagon	111	2	19882.5
7	USA	Jeep	SUV	111	3	24518.333333
8	USA	Ford	Ford Total	110	23	24015.869565
9	USA	USA Total		100	26	24073.846154
10	Grand Total			000	28	24222.071429

## Desired output:

```
proc tabulate data=table format=6.;
class origin make type
  /order=data missing;
var N msrp_mean;
  /* N and msrp_mean were calculated
  in the previous tabulate so just show it
  */
tables origin*make*type
  , N*sum
  msrp_mean*sum*f=dollar7.
;
Run;
```

- Input data set is our summary table
- Order=data – don't re-sort data, it's in the order we want with totals at the bottom
- Missing keyword is required because subtotal subcategories are blank
- Msrp is now msrp\_mean
- All keywords removed – table already has subtotals
- <var>\*sum – data has already been summarised, so just the display the value (sum of itself)

			N	MSRP_Mean
			Sum	Sum
origin	make	type		
Asia	Isuzu	SUV	2	\$26,149
USA	Ford	SUV	4	\$32,055
		Sedan	11	\$20,835
		Sports	3	\$28,418
		Truck	3	\$23,312
		Wagon	2	\$19,883
		Ford Total	23	\$24,018
	Jeep	SUV	3	\$24,518
	USA Total		26	\$24,074
Grand Total			28	\$24,222

## Conclusion

- Most reporting procedures in SAS have OUT= option
- If you don't like the standard output, you can change it!
- Easy to generalize the example above as a macro
- Source code is in the conference proceedings



# SAS<sup>®</sup> GLOBAL FORUM 2018

April 8 - 11 | Denver, CO  
Colorado Convention Center

#SASGF